Causal mediation analysis using Stata

Joerg Luedicke

Senior Social Scientist and Software Developer

StataCorp

May 9, 2024
Florence

Causal inference and mediation

- The goal of causal inference is to identify and quantify the effect of some treatment (exposure) on some outcome.
- With causal mediation, we can disentangle causal effects into direct and indirect effects.
- By decomposing causal effects into direct and indirect effects, we are targeting the underlying mechanism of causal relations.
- In other words, we use causal mediation to learn about *why* certain causes have the effect they have.

Causal diagram illustration

- Simple causal model for the effect of T on Y:



- Causal model for the effect of T on Y through M:



- Mediation model with a direct and an indirect effect:



Potential outcomes framework (1)

- Consider a simple randomized experiment with binary treatment T and outcome Y, with sample observations $i = 1...N$.
- We wish to identify two sets of potential outcomes, $Y_i(1)$ and $Y_i(0)$, where $Y_i(t)$ is the outcome that would be realized if the ith individual were exposed to treatment level t.
- If it were possible to observe an individual in both states at the same time, we would observe one outcome value under treatment, $Y_i(1)$, and one value under the control condition, $Y_i(0)$.
- The (individual-level) treatment effect would then be the difference $\tau_i = Y_i(1) - Y_i(0)$.
- Averaging the difference over all individuals in the sample would yield an estimate of the ATE $\tau = E[Y_i(1) - Y_i(0)] = E[Y_i(1)] - E[Y_i(0)]$.

Potential outcomes framework (2)

- As we know, it is not possible to observe the same individual under both conditions at the same time.
- We can only observe one of these while the other is missing.
- If an individual is treated, we observe $Y_i(1)$, and if not, we observe $Y_i(0)$.
- This has been coined the "fundamental problem of causal inference"
- Much of the treatment effects and causal inference literature deals with the question of how to estimate an ATE in the presence of this problem.
- In a simple experiment where treatment is randomly assigned, the potential outcomes are independent of treatment assignment and the missing potential outcomes are missing completely at random.
- With observational rather than experimental data the potential outcomes are not independent of the treatment assignment process, and the causal effect is not identifiable without imposing further assumptions such as conditional independence.

Potential outcome framework for causal mediation

- Now consider the case where we have a mediator M in addition to treatment T and outcome Y.
- We now have an additional set of potential outcomes, $M_i(1)$ and $M_i(0)$, because M is also causally related to the treatment.
- $M_i(1)$ are the potential outcomes of the mediator that would be observed had the ith individual been assigned to the group of active treatment.
- $M_i(0)$ are the potential outcomes of the mediator that would be observed had the ith individual been assigned to the control group.
- Let t be the treatment level with respect to the outcome, and let t' be the treatment level with respect to the mediator, the potential outcomes become $Y_i[t, M_i(t')]$.

Causal mediation potential outcomes (1)

- With binary treatment, we now have four sets of potential outcomes: $Y_i[0, M_i(0)]$, $Y_i[1, M_i(1)]$, $Y_i[1, M_i(0)]$ and $Y_i[0, M_i(1)]$.
- $\boldsymbol{Y_i[0, M_i(0)]}$ is observed if $T_i = 0$.
- $\boldsymbol{Y_i[1, M_i(1)]}$ is observed if $T_i = 1$.
- $\boldsymbol{Y_i[0, M_i(0)]}$ are the potential outcomes that we would observe if nobody in the population received treatment.
- $\boldsymbol{Y_i[1, M_i(1)]}$ are the potential outcomes that we would observe if everybody in the population received treatment.
- Notice that $Y_i[0, M_i(0)] = Y_i(0)$ and $Y_i[1, M_i(1)] = Y_i(1)$

Causal mediation potential outcomes (2)

- $\boldsymbol{Y_i}[\mathbf{1}, \boldsymbol{M_i}(\mathbf{0})]$ and $\boldsymbol{Y_i}[\mathbf{0}, \boldsymbol{M_i}(\mathbf{1})]$, sometimes referred to as cross-world potential outcomes, are never observed.
- $\boldsymbol{Y_i}[\mathbf{1}, \boldsymbol{M_i}(\mathbf{0})]$ are the potential outcomes that we would observe if everybody in the population received treatment, but where the mediator is held at a value that would be observed as though nobody in the population received treatment.
- $\boldsymbol{Y_i}[\mathbf{0}, \boldsymbol{M_i}(\mathbf{1})]$ are the potential outcomes that we would observe if nobody in the population received treatment, but where the mediator is held at a value that would be observed as though everybody in the population received treatment.

Direct, indirect, and total treatment effects

- Average direct, indirect, and total treatment effects are contrasts between potential outcome means.
- The total effect is:
$$\tau \equiv E[Y_i(1)] - E[Y_i(0)] = E[Y_i(1, M_i(1))] - E[Y_i(0, M_i(0))]$$
- The effect of the treatment on the outcome through the mediator is the indirect effect:
$$\delta(t) \equiv E[Y_i(t, M_i(1))] - E[Y_i(t, M_i(0))], \quad t \in \{0, 1\}$$
- The direct effect of the treatment is:
$$\zeta(t) \equiv E[Y_i(1, M_i(t))] - E[Y_i(0, M_i(t))], \quad t \in \{0, 1\}$$
- Notice that the total effect is the sum of direct and indirect effects
$$\tau = \delta(t) + \zeta(t)$$

Two treatment effect decompositions

- If we include a treatment-mediator interaction, the total treatment effect can be decomposed in two different ways.
- We can decompose the total effect using components
  $\delta(0) \equiv E[Y_i(0, M_i(1))] - E[Y_i(0, M_i(0))]$ and
  $\zeta(0) \equiv E[Y_i(1, M_i(0))] - E[Y_i(0, M_i(0))]$
- ... as well as
  $\delta(1) \equiv E[Y_i(1, M_i(1))] - E[Y_i(1, M_i(0))]$ and
  $\zeta(1) \equiv E[Y_i(1, M_i(1))] - E[Y_i(0, M_i(1))]$
- If we do not include a treatment-mediator interaction, i.e., we impose the assumption that the effect of the mediator on the outcome does not vary across treatment groups, we have that $\delta(0) = \delta(1)$ and $\zeta(0) = \zeta(1)$.

Estimands

- Denoting $E[Y_i(t, M_i(t'))]$ as $Y_{tM_{t'}}$, we define the following treatment effects of interest

(Total) natural indirect effect (**NIE**)	$Y_{1M_1} - Y_{1M_0}$
(Pure) natural direct effect (**NDE**)	$Y_{1M_0} - Y_{0M_0}$
(Pure) natural indirect effect (**PNIE**)	$Y_{0M_1} - Y_{0M_0}$
(Total) natural direct effect (**TNDE**)	$Y_{1M_1} - Y_{0M_1}$
Total effect (**TE**)	$Y_{1M_1} - Y_{0M_0}$

How to identify potential outcome means?

- The potential-outcome means are the result of an integral of the conditional expectation of the outcome with respect to the conditional distribution of the mediator:

$$f[Y_i(t, M_i(t'))|\boldsymbol{X_i} = \boldsymbol{x}] = \int f[Y_i|M_i = m, T_i = t, \boldsymbol{X_i} = \boldsymbol{x}]\, dF[m|T_i = t', \boldsymbol{X_i} = \boldsymbol{x}]$$

- This is sometimes referred to as the "mediation formula".
- It expresses the potential outcomes as a function of the conditional distribution of M_i given T_i and $\boldsymbol{X}_i$, and that of Y_i given M_i, T_i, and $\boldsymbol{X}_i$.
- Notice that this is a nonparametric identification result.

Illustrative example using a linear model (1)

- Suppose we have the following model with two equations:

$$Y_i = \beta_0 + \beta_1 M_i + \beta_2 T_i + \epsilon_i$$
$$M_i = \alpha_0 + \alpha_1 T_i + \nu_i$$

- To calculate the natural indirect effect (NIE), we need estimates for potential-outcome means $E[Y_i(1, M_i(1))]$ and $E[Y_i(1, M_i(0))]$.
- With the linear model, we can write the model in reduced form and yield the conditional expectation of outcome Y:

$$\begin{aligned} E[Y_i|M_i, T_i] &= \beta_0 + \beta_1(\alpha_0 + \alpha_1 T_i) + \beta_2 T_i \\ &= \beta_0 + \beta_1\alpha_0 + \beta_1\alpha_1 T_i + \beta_2 T_i \end{aligned}$$

- To obtain the potential-outcome means, we can modify the reduced-form model by replacing M_i with the expectation of M_i that we would observe if T_i had taken on the value t' for every unit in the population:

$$E[Y_i(t, M_i(t'))] = \beta_0 + \beta_1 E[M_i(t')] + \beta_2 t, \quad t \in \{0, 1\}$$

Illustrative example using a linear model (2)

- Now, to compute the potential-outcome mean $E[Y_i(1, M_i(1))]$, we must set the treatment T_i to 1 in both the outcome and the mediator equations. In other words, we fix both t and t' at 1:

$$
\begin{aligned}
E[Y_i(1, M_i(1))] &= \beta_0 + \beta_1 E[M_i(t')] + \beta_2 t, \quad t = t' = 1 \\
&= \beta_0 + \beta_1\alpha_0 + \beta_1\alpha_1 \times 1 + \beta_2 \times 1 \\
&= \beta_0 + \beta_1\alpha_0 + \beta_1\alpha_1 + \beta_2
\end{aligned}
$$

- To compute $E[Y_i(1, M_i(0))]$, we need to set treatment T_i to 1 in the outcome equation but set it to 0 in the mediator equation. Specifically, we fix $t' = 0$ and $t = 1$:

$$
\begin{aligned}
E[Y_i(1, M_i(0))] &= \beta_0 + \beta_1 E[M_i(t')] + \beta_2 t, \quad t = 1;\ t' = 0 \\
&= \beta_0 + \beta_1\alpha_0 + \beta_1\alpha_1 \times 0 + \beta_2 \times 1 \\
&= \beta_0 + \beta_1\alpha_0 + \beta_2
\end{aligned}
$$

Illustrative example using a linear model (3)

- Calculating the difference yields the indirect treatment effect

$$\begin{aligned}\delta(1) &= (\beta_0 + \beta_1\alpha_0 + \beta_1\alpha_1 + \beta_2) - (\beta_0 + \beta_1\alpha_0 + \beta_2) \\ &= \beta_0 + \beta_1\alpha_0 + \beta_1\alpha_1 + \beta_2 - \beta_0 - \beta_1\alpha_0 - \beta_2 \\ &= \beta_1\alpha_1\end{aligned}$$

- We are left with the product of the treatment coefficient from the mediator equation and the mediator coefficient from the outcome equation.
- This is congruent with the classical product-of-coefficients method.
- Had we included a treatment-mediator interaction, the result would be $\delta(1) = (\beta_1 + \beta_3)\alpha_1$.
- Not as simple for models other than the linear model.

Assumptions for identifying estimands of interest

- SUTVA, overlap, **sequential ignorability**
- Sequential ignorability essentially means
  - No unobserved confounding in the treatment-outcome relationship.
  - No unobserved confounding in the mediator-outcome relationship.
  - No unmeasured confounding in the treatment-mediator relationship.
  - There are no (observed) confounders in the mediator-outcome relationship that are caused by the treatment.

Stata's `mediate` command

- New in Stata 18: `mediate`
- `mediate` performs causal mediation analysis for linear and generalized linear models.
- It uses analytical expressions to compute potential outcome means based on parametric models.
- Outcome and mediator variables may be continuous, binary, or count.
- Treatment may be binary, multivalued, or continuous.
- Linear, logit, probit, Poisson, and exponential-mean models for outcome and mediator.
- Special-purpose postestimation commands.

Outcome and mediator model combinations

	linear	logit	probit	Poisson	exp. mean
linear	x	x	x	x	x
logit		x	x	x	
probit	x	x	x	x	x
Poisson	x	x	x	x	x
exp. mean	x	x	x	x	x

Note: x indicates supported model combination

Postestimation commands

- Special-purpose postestimation commands include
  - `estat proportion` proportion mediated
  - `estat cde` controlled direct effects
  - `estat rr` treatment effects as risk ratios
  - `estat or` treatment effects as odds ratios
  - `estat irr` treatment effects as incidence-rate ratios
  - `estat effectsplot` plot treatment effects

Treatment effects on different scales (1)

- If the outcome is binary, and if the outcome model is either `logit` or `probit`, we can express the treatment effects as risk ratios or odds ratios.
- If the outcome model is Poisson/exponential mean, treatment effects can be expressed as incidence-rate ratios.
- The treatment effects on risk-ratio and incidence-rate-ratio scales are ratios of potential-outcome means:

$$
\begin{aligned}
\text{NIE}^{\text{RR}} &\equiv Y_{1M_1} / Y_{1M_0} \\
\text{NDE}^{\text{RR}} &\equiv Y_{1M_0} / Y_{0M_0} \\
\text{PNIE}^{\text{RR}} &\equiv Y_{0M_1} / Y_{0M_0} \\
\text{TNDE}^{\text{RR}} &\equiv Y_{1M_1} / Y_{0M_1} \\
\text{TE}^{\text{RR}} &\equiv Y_{1M_1} / Y_{0M_0}
\end{aligned}
$$

Treatment effects on different scales (2)

- For logit and probit outcome models, $Y_{tM_{t'}}$ are probabilities, and so the treatment effects on odds-ratio scale are

$$\begin{aligned}
\text{NIE}^{\text{OR}} &\equiv Y_{1M_1}/(1-Y_{1M_1})/\{Y_{1M_0}/(1-Y_{1M_0})\} \\
\text{NDE}^{\text{OR}} &\equiv Y_{1M_0}/(1-Y_{1M_0})/\{Y_{0M_0}/(1-Y_{0M_0})\} \\
\text{PNIE}^{\text{OR}} &\equiv Y_{0M_1}/(1-Y_{0M_1})/\{Y_{0M_0}/(1-Y_{0M_0})\} \\
\text{TNDE}^{\text{OR}} &\equiv Y_{1M_1}/(1-Y_{1M_1})/\{Y_{0M_1}/(1-Y_{0M_1})\} \\
\text{TE}^{\text{OR}} &\equiv Y_{1M_1}/(1-Y_{1M_1})/\{Y_{0M_0}/(1-Y_{0M_0})\}
\end{aligned}$$

- Notice that for all of these scales, the decomposition becomes multiplicative; that is, the total effect becomes the product of direct and indirect effects.

Controlled direct effects

- A controlled direct effect (CDE) is the effect of a treatment on an outcome when the mediator is fixed at a particular value.
- To estimate controlled direct effects, we use only the results of the outcome equation.
- Rather than having potential outcomes of the form $Y_i(t, M_i(t'))$, here we have potential outcomes $Y_i(t|M_i = m)$.
- That is, we have potential outcomes for each treatment level t that are evaluated at value m of the mediator.
- CDE(m) is then the average of the differences between potential outcomes.
- For binary treatment, CDE(m) is defined as $Y_i(1|M_i = m) - Y_i(0|M_i = m)$.
- Letting Y_{tm} be a shorthand for $Y_i(t|M_i = m)$, we have that

$$\begin{aligned}
\text{CDE(m)} &\equiv Y_{1m} - Y_{0m} \\
\text{CDE(m)}^{\text{RR}} &\equiv Y_{1m}/Y_{0m} \\
\text{CDE(m)}^{\text{IRR}} &\equiv Y_{1m}/Y_{0m} \\
\text{CDE(m)}^{\text{OR}} &\equiv Y_{1m}/(1 - Y_{1m})/\{Y_{0m}/(1 - Y_{0m})\}
\end{aligned}$$

Example data (1)

```
. webuse wellbeing
(Fictional well-being data)

. list wellbeing bonotonin exercise age gender in 1/5, abbreviate(12) clean

       wellbeing   bonotonin   exercise   age   gender
  1.    71.73816    196.5467    Control    58     Male
  2.    68.66573    195.8572   Exercise    38   Female
  3.    71.05155    228.6035   Exercise    53   Female
  4.    69.44469    206.6651   Exercise    44   Female
  5.    75.62035    261.6855   Exercise    28   Female
```

Linear models with no treatment-mediator interaction

```
. mediate (wellbeing) (bonotonin) (exercise), nointeraction

Iteration 0:  EE criterion =  6.800e-28
Iteration 1:  EE criterion =  1.777e-28

Causal mediation analysis                            Number of obs = 2,000

Outcome model:     Linear
Mediator model:    Linear
Mediator variable: bonotonin
Treatment type:    Binary
```

wellbeing	Coefficient	Robust std. err.	z	P>|z|	[95% conf.	interval]
NIE						
exercise (Exercise vs Control)	9.694617	.377312	25.69	0.000	8.955099	10.43413
NDE						
exercise (Exercise vs Control)	2.996658	.2109357	14.21	0.000	2.583231	3.410084
TE						
exercise (Exercise vs Control)	12.69127	.4005769	31.68	0.000	11.90616	13.47639

Note: Outcome equation does not include treatment-mediator interaction.

Proportion mediated

```
. estat proportion

Proportion mediated                                    Number of obs = 2,000
```

wellbeing	Proportion	Robust std. err.	z	P>|z|	[95% conf. interval]	
exercise (Exercise vs Control)	.7638805	.0154928	49.31	0.000	.7335151	.7942459

Linear models with treatment-mediator interaction

```
. mediate (wellbeing basewell age gender hstatus)          ///
>         (bonotonin basebono age gender hstatus)          ///
>         (exercise)

Iteration 0:  EE criterion =  2.004e-27
Iteration 1:  EE criterion =  2.804e-28

Causal mediation analysis                            Number of obs = 2,000

Outcome model:      Linear
Mediator model:     Linear
Mediator variable: bonotonin
Treatment type:     Binary
```

wellbeing	Coefficient	Robust std. err.	z	P>|z|	[95% conf.	interval]
NIE						
exercise (Exercise vs Control)	10.02204	.2256812	44.41	0.000	9.579717	10.46437
NDE						
exercise (Exercise vs Control)	3.085412	.168631	18.30	0.000	2.754901	3.415922
TE						
exercise (Exercise vs Control)	13.10746	.2304752	56.87	0.000	12.65573	13.55918

```
Note: Outcome equation includes treatment-mediator interaction.
```

Estimating potential outcome means

```
. mediate (wellbeing basewell age gender hstatus)          ///
>         (bonotonin basebono age gender hstatus)          ///
>         (exercise), pom

Iteration 0:  EE criterion =  2.050e-27
Iteration 1:  EE criterion =  2.775e-28

Causal mediation analysis                             Number of obs = 2,000

Outcome model:      Linear
Mediator model:     Linear
Mediator variable: bonotonin
Treatment type:     Binary
```

wellbeing	Coefficient	Robust std. err.	z	P>|z|	[95% conf.	interval]
POmeans						
Y0M0	56.89975	.228515	249.00	0.000	56.45187	57.34763
Y1M0	59.98516	.2555341	234.74	0.000	59.48432	60.486
Y0M1	66.83246	.2644294	252.74	0.000	66.31419	67.35073
Y1M1	70.0072	.2314185	302.51	0.000	69.55363	70.46077

Note: Outcome equation includes treatment-mediator interaction.

Estimating all effects and potential outcome means

```
. mediate (wellbeing basewell age gender hstatus)          ///
>         (bonotonin basebono age gender hstatus)          ///
>         (exercise), all

Iteration 0:  EE criterion =  2.132e-27
Iteration 1:  EE criterion =  3.527e-28

Causal mediation analysis                              Number of obs = 2,000

Outcome model:     Linear
Mediator model:    Linear
Mediator variable: bonotonin
Treatment type:    Binary
------------------------------------------------------------------------------
             |               Robust
   wellbeing | Coefficient  std. err.      z    P>|z|     [95% conf. interval]
-------------+----------------------------------------------------------------
POmeans      |
        Y0M0 |   56.89975    .228515   249.00   0.000     56.45187    57.34763
        Y1M0 |   59.98516   .2555341   234.74   0.000     59.48432      60.486
        Y0M1 |   66.83246   .2644294   252.74   0.000     66.31419    67.35073
        Y1M1 |    70.0072   .2314185   302.51   0.000     69.55363    70.46077
-------------+----------------------------------------------------------------
NIE          |
    exercise |
  (Exercise  |
         vs  |
   Control)  |   10.02204   .2256812    44.41   0.000     9.579717    10.46437
-------------+----------------------------------------------------------------
NDE          |
    exercise |
  (Exercise  |
         vs  |
   Control)  |   3.085412    .168631    18.30   0.000     2.754901    3.415922
-------------+----------------------------------------------------------------
PNIE         |
    exercise |
  (Exercise  |
         vs  |
   Control)  |   9.932713   .2290178    43.37   0.000     9.483846    10.38158
-------------+----------------------------------------------------------------
TNDE         |
    exercise |
  (Exercise  |
         vs  |
   Control)  |   3.174743   .1808011    17.56   0.000     2.820379    3.529107
-------------+----------------------------------------------------------------
TE           |
    exercise |
  (Exercise  |
         vs  |
   Control)  |   13.10746   .2304752    56.87   0.000     12.65573    13.55918
------------------------------------------------------------------------------
Note: Outcome equation includes treatment-mediator interaction.
```

Auxiliary parameters

```
. mediate, aequations

Causal mediation analysis                              Number of obs = 2,000

Outcome model:     Linear
Mediator model:    Linear
Mediator variable: bonotonin
Treatment type:    Binary

------------------------------------------------------------------------------
             |               Robust
   wellbeing | Coefficient  std. err.      z    P>|z|     [95% conf. interval]
-------------+----------------------------------------------------------------
POmeans      |
        Y0M0 |   56.89975    .228515   249.00   0.000     56.45187    57.34763
        Y1M0 |   59.98516   .2555341   234.74   0.000     59.48432      60.486
        Y0M1 |   66.83246   .2644294   252.74   0.000     66.31419    67.35073
        Y1M1 |    70.0072   .2314185   302.51   0.000     69.55363    70.46077
-------------+----------------------------------------------------------------
NIE          |
    exercise |
  (Exercise  |
         vs  |
   Control)  |   10.02204   .2256812    44.41   0.000     9.579717    10.46437
-------------+----------------------------------------------------------------
NDE          |
    exercise |
  (Exercise  |
         vs  |
   Control)  |   3.085412    .168631    18.30   0.000     2.754901    3.415922
-------------+----------------------------------------------------------------
PNIE         |
    exercise |
  (Exercise  |
         vs  |
   Control)  |   9.932713   .2290178    43.37   0.000     9.483846    10.38158
-------------+----------------------------------------------------------------
TNDE         |
    exercise |
  (Exercise  |
         vs  |
   Control)  |   3.174743   .1808011    17.56   0.000     2.820379    3.529107
-------------+----------------------------------------------------------------
TE           |
    exercise |
  (Exercise  |
         vs  |
   Control)  |   13.10746   .2304752    56.87   0.000     12.65573    13.55918
-------------+----------------------------------------------------------------
wellbeing    |
    exercise |
   Exercise  |   2.777685   .6449446     4.31   0.000     1.513616    4.041753
   bonotonin |   .2141319   .0026418    81.05   0.000      .208954    .2193099
             |
    exercise#|
 c.bonotonin |
   Exercise  |   .0019258   .0034941     0.55   0.582    -.0049224    .0087741
             |
    basewell |   .1685634   .0038294    44.02   0.000     .1610579    .1760689
         age |   .0266714   .0072856     3.66   0.000      .012392    .0409509
      gender |  -.1031899   .1282411    -0.80   0.421    -.3545379    .1481581
     hstatus |   .9787586   .0773014    12.66   0.000     .8272506    1.130267
       _cons |   9.508461   .5918832    16.06   0.000     8.348391    10.66853
-------------+----------------------------------------------------------------
bonotonin    |
    exercise |
   Exercise  |   46.38595   .8963335    51.75   0.000     44.62916    48.14273
    basebono |   1.019825   .0154741    65.91   0.000     .9894966    1.050154
         age |    .368765    .049656     7.43   0.000      .271441     .466089
      gender |   5.648102   .8946093     6.31   0.000       3.8947    7.401504
     hstatus |   3.662454   .5353618     6.84   0.000     2.613164    4.711743
       _cons |  -40.92391   3.646124   -11.22   0.000    -48.07018   -33.77764
------------------------------------------------------------------------------
Note: Outcome equation includes treatment-mediator interaction.
```

Binary outcome and mediator

```
. mediate (bwellbeing basewell age gender hstatus, logit)        ///
>         (bbonotonin basebono age gender hstatus, logit)        ///
>         (exercise), nointeraction

Iteration 0:  EE criterion =  1.413e-18
Iteration 1:  EE criterion =  1.371e-32

Causal mediation analysis                             Number of obs = 2,000

Outcome model:      Logit
Mediator model:     Logit
Mediator variable: bbonotonin
Treatment type:     Binary
```

bwellbeing	Coefficient	Robust std. err.	z	P>|z|	[95% conf. interval]	
NIE						
exercise (Exercise vs Control)	.1053001	.0142631	7.38	0.000	.0773449	.1332553
NDE						
exercise (Exercise vs Control)	.1528838	.0189013	8.09	0.000	.115838	.1899296
TE						
exercise (Exercise vs Control)	.2581839	.014312	18.04	0.000	.2301328	.286235

Note: Outcome equation does not include treatment-mediator interaction.

Treatment effects as risk ratios

```
. estat rr
estat rr requires potential-outcome means; refitting model ...

Transformed treatment effects                          Number of obs = 2,000
```

bwellbeing	Risk ratio	Robust std. err.	z	P>|z|	[95% conf. interval]	
NIE						
exercise (Exercise vs Control)	1.22985	.0383193	6.64	0.000	1.156993	1.307295
NDE						
exercise (Exercise vs Control)	1.500861	.0714322	8.53	0.000	1.367188	1.647603
TE						
exercise (Exercise vs Control)	1.845833	.0706637	16.01	0.000	1.712403	1.98966

Treatment effects as odds ratios

```
. estat or
estat or requires potential-outcome means; refitting model ...

Transformed treatment effects                          Number of obs = 2,000
```

bwellbeing	Odds ratio	Robust std. err.	z	P>|z|	[95% conf.	interval]
NIE						
exercise (Exercise vs Control)	1.526485	.087768	7.36	0.000	1.363802	1.708575
NDE						
exercise (Exercise vs Control)	1.924312	.1529157	8.24	0.000	1.646777	2.248621
TE						
exercise (Exercise vs Control)	2.937434	.1841548	17.19	0.000	2.597791	3.321482

Example data (2)

```
. webuse birthweight
(Fictional birthweight data)

. list bweight ncigs college ses sespar age in 1/5, clean

       bweight   ncigs   college        ses     sespar   age
  1.      3621       1        No     5.3581   3.308523    29
  2.      3278       0       Yes   9.556957   4.376035    38
  3.      3073       1        No   3.980829   6.580275    39
  4.      3306       0       Yes   11.17643   12.12075    30
  5.      4517       0       Yes   9.026146   4.738766    28
```

Exponential mean and Poisson models

```
. mediate (bweight sespar c.age##c.age, expmean) ///
>         (ncigs sespar c.age##c.age, poisson) ///
>         (college), nointeract

Iteration 0:  EE criterion =  3.250e-13
Iteration 1:  EE criterion =  9.147e-18

Causal mediation analysis                               Number of obs = 2,000

Outcome model:      Exponential mean
Mediator model:     Poisson
Mediator variable:  ncigs
Treatment type:     Binary
```

bweight	Coefficient	Robust std. err.	z	P>|z|	[95% conf.	interval]
NIE						
college						
(Yes vs No)	198.978	23.53279	8.46	0.000	152.8546	245.1014
NDE						
college						
(Yes vs No)	320.3318	34.47792	9.29	0.000	252.7563	387.9072
TE						
college						
(Yes vs No)	519.3098	28.70435	18.09	0.000	463.0503	575.5693

Note: Outcome equation does not include treatment-mediator interaction.

Estimating incidence-rate ratios

```
. estat irr
estat irr requires potential-outcome means; refitting model ...

Transformed treatment effects                          Number of obs = 2,000
```

bweight	IRR	Robust std. err.	z	P>|z|	[95% conf.	interval]
NIE						
college						
(Yes vs No)	1.057819	.0072037	8.25	0.000	1.043794	1.072033
NDE						
college						
(Yes vs No)	1.102636	.0113921	9.46	0.000	1.080533	1.125192
TE						
college						
(Yes vs No)	1.16639	.009948	18.05	0.000	1.147055	1.186052

Estimating controlled direct effects

```
. estat cde, mvalue(0 1)

Controlled direct effect                              Number of obs = 2,000

Mediator variable: ncigs
Mediator values:
  1._at: ncigs = 0
  2._at: ncigs = 1
```

	CDE	Delta-method std. err.	z	P>|z|	[95% conf.	interval]
college@_at						
(Yes vs No) 1	341.955	35.26807	9.70	0.000	272.8308	411.0791
(Yes vs No) 2	332.6419	34.94916	9.52	0.000	264.1428	401.141

Estimating differences between controlled direct effects

```
. estat cde, mvalue(0 1) contrast

Controlled direct effect                              Number of obs = 2,000

Mediator variable: ncigs
Mediator values:
  1._at: ncigs = 0
  2._at: ncigs = 1

------------------------------------------------------------------------------
             |            Delta-method
             |        CDE   std. err.      z    P>|z|     [95% conf. interval]
-------------+----------------------------------------------------------------
 _at#college |
   (2 vs 1)  |
 (Yes vs No) |  -9.313066   .9748033    -9.55   0.000    -11.22365   -7.402487
------------------------------------------------------------------------------
```

More treatment interactions

```
. mediate (bweight sespar c.age##c.age                              ///
>                 i.college#(c.sespar c.age##c.age), expmean)     ///
>         (ncigs c.sespar c.age##c.age                              ///
>                 i.college#(c.sespar c.age##c.age), poisson)     ///
>         (college)

Iteration 0:  EE criterion =  1.691e-12
Iteration 1:  EE criterion =  1.122e-14

Causal mediation analysis                          Number of obs = 2,000

Outcome model:      Exponential mean
Mediator model:     Poisson
Mediator variable: ncigs
Treatment type:     Binary
```

bweight	Coefficient	Robust std. err.	z	P>|z|	[95% conf.	interval]
NIE						
college						
(Yes vs No)	111.6007	67.53715	1.65	0.098	-20.76971	243.971
NDE						
college						
(Yes vs No)	407.5962	72.49614	5.62	0.000	265.5063	549.686
TE						
college						
(Yes vs No)	519.1968	28.71853	18.08	0.000	462.9095	575.4841

Note: Outcome equation includes treatment-mediator interaction.

Mediator interactions

```
. mediate (bweight sespar c.age##c.age                          ///
>                  c.ncigs#(c.sespar c.age##c.age), expmean)     ///
>           (ncigs c.sespar c.age##c.age, poisson)               ///
>           (college)

Iteration 0:  EE criterion =  6.460e-11
Iteration 1:  EE criterion =  5.837e-15

Causal mediation analysis                          Number of obs = 2,000

Outcome model:     Exponential mean
Mediator model:    Poisson
Mediator variable: ncigs
Treatment type:    Binary
```

bweight	Coefficient	Robust std. err.	z	P>|z|	[95% conf.	interval]
NIE						
college						
(Yes vs No)	86.26849	68.96645	1.25	0.211	-48.90328	221.4403
NDE						
college						
(Yes vs No)	431.5822	73.70305	5.86	0.000	287.1269	576.0375
TE						
college						
(Yes vs No)	517.8507	28.64809	18.08	0.000	461.7015	573.9999

Note: Outcome equation includes treatment-mediator interaction.

Multivalued treatment

```
. mediate (wellbeing age gender i.hstatus basewell)       ///
>         (bonotonin basebono)                            ///
>         (mexercise)

Iteration 0:  EE criterion =  1.778e-27
Iteration 1:  EE criterion =  1.198e-27

Causal mediation analysis                                Number of obs = 2,000

Outcome model:     Linear
Mediator model:    Linear
Mediator variable: bonotonin
Treatment type:    Multivalued
```

wellbeing	Coefficient	Robust std. err.	z	P>|z|	[95% conf.	interval]
NIE						
mexercise						
(45 minutes vs Control)	5.128899	.3505171	14.63	0.000	4.441898	5.815899
(90 minutes vs Control)	9.780537	.2880877	33.95	0.000	9.215895	10.34518
NDE						
mexercise						
(45 minutes vs Control)	1.197498	.1750038	6.84	0.000	.8544965	1.540499
(90 minutes vs Control)	3.051084	.2071236	14.73	0.000	2.645129	3.457039
TE						
mexercise						
(45 minutes vs Control)	6.326396	.3894269	16.25	0.000	5.563134	7.089659
(90 minutes vs Control)	12.83162	.2967962	43.23	0.000	12.24991	13.41333

Note: Outcome equation includes treatment-mediator interaction.

Continuous treatment

```
. webuse birthweight
(Fictional birthweight data)

. qui sum ses

. generate std_ses = (ses-r(mean))/r(sd)

. mediate (bweight sespar c.age##c.age, expmean)  ///
>         (ncigs sespar c.age##c.age, poisson)    ///
>         (std_ses, continuous(0 2)), nointeract

Iteration 0:  EE criterion =  1.470e-12
Iteration 1:  EE criterion =  1.816e-17

Causal mediation analysis                              Number of obs = 2,000

Outcome model:      Exponential mean
Mediator model:     Poisson
Mediator variable: ncigs
Treatment type:     Continuous

Continuous treatment levels:
  0: std_ses = 0 (control)
  1: std_ses = 2
```

bweight	Coefficient	Robust std. err.	z	P>|z|	[95% conf.	interval]
NIE						
std_ses (1 vs 0)	110.1346	8.724232	12.62	0.000	93.03538	127.2337
NDE						
std_ses (1 vs 0)	180.0172	34.77372	5.18	0.000	111.8619	248.1724
TE						
std_ses (1 vs 0)	290.1517	33.85571	8.57	0.000	223.7958	356.5077

Note: Outcome equation does not include treatment-mediator interaction.

Continuous treatment with multiple evaluation points

```
. mediate (bweight sespar c.age##c.age, expmean)  ///
>         (ncigs sespar c.age##c.age, poisson)    ///
>         (std_ses, continuous(0 -2 -1 1 2)), nointeract

Iteration 0:  EE criterion =  1.470e-12
Iteration 1:  EE criterion =  2.374e-17

Causal mediation analysis                                 Number of obs = 2,000

Outcome model:     Exponential mean
Mediator model:    Poisson
Mediator variable: ncigs
Treatment type:    Continuous

Continuous treatment levels:
  0: std_ses =  0 (control)
  1: std_ses = -2
  2: std_ses = -1
  3: std_ses =  1
  4: std_ses =  2

------------------------------------------------------------------------------
             |               Robust
     bweight | Coefficient  std. err.      z    P>|z|     [95% conf. interval]
-------------+----------------------------------------------------------------
NIE          |
     std_ses |
   (1 vs 0)  |  -276.2757   27.69004    -9.98   0.000    -330.5471   -222.0042
   (2 vs 0)  |  -100.1155   9.170566   -10.92   0.000    -118.0894   -82.14148
   (3 vs 0)  |   65.84585   5.423096    12.14   0.000     55.21678    76.47493
   (4 vs 0)  |   110.1346   8.724232    12.62   0.000     93.03538    127.2337
-------------+----------------------------------------------------------------
NDE          |
     std_ses |
   (1 vs 0)  |  -170.9012   31.33649    -5.45   0.000    -232.3196   -109.4828
   (2 vs 0)  |  -86.56069   16.08129    -5.38   0.000    -118.0794   -55.04193
   (3 vs 0)  |   88.83929   16.94031     5.24   0.000      55.6369    122.0417
   (4 vs 0)  |   180.0172   34.77372     5.18   0.000     111.8619    248.1724
-------------+----------------------------------------------------------------
TE           |
     std_ses |
   (1 vs 0)  |  -447.1769   35.41401   -12.63   0.000    -516.5871   -377.7667
   (2 vs 0)  |  -186.6761   15.73291   -11.87   0.000    -217.5121   -155.8402
   (3 vs 0)  |   154.6851   16.31969     9.48   0.000     122.6991    186.6712
   (4 vs 0)  |   290.1517   33.85571     8.57   0.000     223.7958    356.5077
------------------------------------------------------------------------------
Note: Outcome equation does not include treatment-mediator interaction.

. estat effectplot
```

Plotting treatment effects



Final remarks

- Learn more:
  `https://www.stata.com/manuals/causalmediate.pdf`

Thank you!